

A-874

NORTH DAKOTA STATE UNIVERSITY FEBRUARY 1986

NORTH DAKOTA
STATE DEPOSITORY
DOCUMENT

JUN 02 1986


SERIALS DEPT.
LIBRARY

WILDOAT



3 0109 00746 4368

Wheat and Barley Yield Reductions Due to Wild Oat Infestations



Version 1.1
for the
IBM Personal Computer

By

E.H. Vasey
Extension Soils Specialist

J.D. Nalewaja
Professor of Weed Science, Agronomy Department

Mike Vasey and Rick Koon
Student Computer Programmers

Plant Science Section
North Dakota Cooperative Extension Service
North Dakota State University



Introduction

North Dakota producers likely spend more for control of wildoat infestations in small grains than for any other weed. Losses due to wildoat infestations in small grains have been estimated at \$130 million annually.

"WILDOAT" is designed to predict yield reduction due to wildoat infestation in either wheat or barley. "WILDOAT" will also allow calculation of cost versus return for recommended herbicide applications. "WILDOAT" was written in BASICA and uses a database built with PC-FILE. "WILDOAT" was developed by Dr. E.H. Vasey, Extension Soils Specialist at North Dakota State University using data collected by Dr. John Nalewaja, Agronomy Department, North Dakota State University (1,2). Programming was done by Michael Vasey and Rick Koon, Student programmers, North Dakota State University.

Instructions

Insert the "WILDOAT" diskette in drive A and turn your system on. The program will automatically load and run. If your system is on, insert the diskette into drive A and re-boot the system by simultaneously depressing the Ctrl and Alt keys, then press the Del key.

Screen 1: Presents an overview of the program and references Extension Circulars pertinent to the program ("Agricultural Weed Control Guide" [3] and "Crop Production Guide" [4]).

Screen 2: Presents a menu for crop selection and a prompt for identification of crop choice by number. You may also seek help or stop (return to DOS) the program by pressing the indicated function keys on line 25 of the screen.

Screen 3: You are first prompted for the projected yield of the selected crop in bushels per acre (bu/A). Normal yield estimates can be based on experience, on variety performance data from branch experiment stations or on detailed soil survey reports. Enter a yield rounded to the nearest whole number or no more than one decimal point to the right (e.g. 34 or 34.5).

If barley is selected, you will be asked about the soil fertility level by indicating if the barley is to be grown on a soil with high fertility (fertilized) or low fertility (non-fertilized). The data used in the program showed less yield loss due to wildoat competition when the barley was grown under high fertility (fertilized) (2). The difference due to soil fertility level (fertilization) of wheat did not justify using separate equations (1).

You will then be prompted for the level of infestation of wildoat plants per square yard. A good estimate can be obtained by counting the wildoat plants in several 1 square yard samples and finding the average. Some rough estimates might be 10 for a mild infestation, 40 for a moderate infestation and 80 or more for a severe infestation. Enter an infestation level rounded to the nearest whole number.

In either case you may request help or stop the program (return to DOS) by pressing the appropriate function keys (NOTE: this will hold true throughout the program when the keys are displayed on line 25 of the screen and will not be mentioned again in this guide).

Screen 4: Presents the predicted percent yield reduction and the expected yield after the percent yield reduction has been applied to the predicted yield. You then have the option of seeing a cost versus return routine. Answering yes presents screen 5.

Screen 5: Screen 5 begins by displaying all recommended herbicides for the chosen crop. A recommended herbicide can be selected by number or a herbicide of your choice can be inputted by pressing the [F1] key. You will then be prompted to answer 4 additional questions: 1) cost per acre for application—this can be your inputted cost or the cost for typical custom application rates; 2) price per gallon or per pound for the herbicide. If you purchase herbicides in large quantities and need to find a per gallon or per pound price the program will assist if the HELP function ([F1] key) is selected. Here you only need to know the size of the container (i.e. 30 gal drum, 5 gal bucket, 100 lb sack) and the cost for the container. The program will calculate the price and automatically store it for use; 3) you will then be prompted for the application rate in pints per acre or pounds per acre. You should make sure to enter the appropriate figures since inaccurate figures will yield false results! The low and high recommended rates are provided for the recommended herbicides and 4) the last prompt in this series asks for the expected selling price per bushel for the chosen crop. Any price can be entered but realistic prices will provide more accurate results than unrealistic prices. Daily price quotes from news sources, quotes from an elevator or prices based on experience would all be good sources.

The program will display the given information and ask if the data is correct after all the information has been entered. Responding yes will clear the screen and present the user with cost, control and return figures and a statement on whether wildoat control would or would not be economical. The results may be printed for a permanent record. (Make sure your printer is on!) If you respond no to the data correct question, you will be prompted to give the number of the statement with the incorrect data. The question corresponding to the given number will appear and the error can be corrected. This option can be continued as long as needed to update or correct the information.

After responding to the printout option, you will be prompted to compare cost versus return for another herbicide. Responding yes returns the program to screen 5 and the prompts repeat from there. Answering no clears the screen and asks if you would like to make another prediction. Answering no stops program execution and returns control to DOS, responding yes returns the program to screen 2.

A note for WILDOAT users

Wildoat 1.1 uses a database built by PC-File. You may add, delete or modify the records for the recommended herbicides by running PC-File in drive A and using the wildoat disk in drive B. The database is labeled WILDOATS and consists of seven fields. 1) Herbicide—the commercial name of the herbicide, 2) crop—what crop the herbicide is used on (NOTE: if the herbicide can be used on wheat and barley you will need two entries, one for wheat and one for barley), 3) low rate—the recommended low rate of application as listed in the latest edition of 'Agricultural Weed Control Guide' (3), 4) high rate—the recommended high rate of application as listed in the latest edition of 'Agricultural Weed Con-

trol Guide' (3), 5) costs—the cost per gallon or per pound for the herbicide, 6) rating—the relative herbicide effectiveness as indicated in the 'Agricultural Weed Control Guide' (3) and 7) comments—any remarks you would like to make up to 35 characters in length.

By using a database approach you can modify the data to fit local conditions.

REFERENCES

1. Bell, A.R. and Nalewaja, J.D., 1968, "Competition of wild oats in wheat and barley." *Weed Sci.* 16:505-508.
2. Dew, D.A., 1972, "An Index of Competition for Estimating Crop Loss Due to Weeds," *Canadian Journal of Plant Science* 52:921-927.
3. Eberlein, C.V., editor, 1984, "Agricultural Weed Control Guide."
4. North Dakota Agricultural Association, "The 1984 Crop Production Guide."

North Dakota State University and the program authors shall have no liability or responsibility to any person or entity with respect to any liability, loss or damage caused or alleged to be caused directly or indirectly by this software, including, but not limited to any interruption of service, loss of business or anticipatory profits or consequential damages resulting from the use or operation of this software.

WILDOAT is designed to predict yield reduction due to wild oat infestation in either wheat or barley.

WILDOAT will also allow you to calculate cost vs. return for recommended herbicide application.

WILDOAT was developed by Dr. E.H. Vasey, Extension Soils Specialist at North Dakota State University.

WILDOAT utilizes data collected by the Weed Science Staff in the Agronomy Department at NDSU.

Refer to current Weed Control Guide (Extension Circular W-253) or current Crop Production Guide for more complete information on wild oat control in wheat or barley.

press any key to continue

SCREEN 1

CROP CHOICES ARE:

1. WHEAT
2. BARLEY

Enter your choice of crop (by number)

[F1] HELP [F2] STOP

SCREEN 2

What is your projected yield of BARLEY in bu/a 33

Is your BARLEY on fertilized (f) or non-fertilized (nf) soil (fertilized consists of adequate rates of Nitrogen and Phosphorus) f

What is the number of wildoat plants per square yard (if the number exceeds 250 then use 250) 100

[F1] HELP [F2] STOP

SCREEN 3 (after selecting Barley option)

An infestation of 100 wildoats per square yard can reduce your yield by 16 percent.

Your yield may be reduced from 33 bu/a to 27.7 bu/a unless wild oats are controlled.

Do you wish to see a cost vs. return for herbicide application (Y/N)

SCREEN 4

The recommended herbicides for BARLEY are:

HERBICIDE	APPLICATION RATES/A			RATING	COMMENTS
	LOW	HIGH	COST		
1 AVADEX (L)	1.25pt	1.25pt	32.00/gal	G	
2 AVENGE	2.5pt	4.0pt	36.50/gal	G	
3 CARBYNE	1.0pt	1.5pt	39.50/gal	F-G	Control may be erratic see label
4 FAR-GO 10G	10 lb	15 lb	0.71/gal	G	for spring-fall rate
5 FAR-GO 4L	1.25pt	1.25pt	32.00/gal	G	
6 HOELON	2.0pt	2.7pt	46.00/gal	G	Foxtail control also


RATINGS: G = GOOD F = FAIR P = POOR N = NONE S = SELDOM O = OFTEN

1. Choose by number the herbicide you will apply ([F1] for own)

[F1] Own herbicide

SCREEN 5 (shows the first of five prompts)

BASIC PROGRAM FLOW FOR WILDOAT



DATA DICTIONARY FOR WILDOAT

ST\$	- A string of 79 blanks. Used throughout the program to erase lines	Price, Pr	- the price the user will pay per gallon or per pound for the herb\$
FKEY\$	- labels the function keys that the user has access to	Rate	- the rate in pints/A or pounds/A herb\$ will be applied
Num%	- the number of records currently stored in WILDOATS.INX	Sell	- the price the user expects to sell his crop at. The price is per bushel.
FILE\$	- a two dimensional array containing NUM% records each consisting of seven fields:	TAL	- total price/A for application and herb\$
	1) Herb\$ = the commercial name of the herbicide	NCR	- gross return a user can expect with no control of wildoats
	2) Crop\$ = the crop (wheat and/or barley) the herbicide is recommended for	CR	- the gross return a user can expect with 100% control of wildoats
	3) Low\$ = the recommended low application rate	D	- difference when compared with TAL will determine if control is cost effective
	4) High\$ = the recommended high application rate	Fert\$	- used in barley routines. Holds values "f" for fertilized soil or "nf" for non-fertilized soil
	5) Rating\$ = the effectiveness of the herbicide	O,W,A,B	- internal variables to hold various results to equations
	6) Comments\$ = any remarks up to 34 characters in length	J%, I%	- used in For-Next loops
		HN	- herbicide number. Allows the user to select herbicide by number rather than typing in the name.
A\$	- used with inkey\$ to allow user to control program continuation	The following variable are from the Screenin routine taken from "Advanced Basic and Beyond" by Dr. L. Goldstein. See program listing or text for definitions.	
Crop	- the number of the crop chosen by the user	XFLD	Csr
Number, Number\$	- the number of wildoats (user input)	YFLD	Inputend
PYR%	- percent yield reduction, based on equations drived by Weed Science Department, NDSU	S\$	true
		T\$	false
		Lastpos	C
		caller	C\$
		lngth	E\$
YR	- yield reduction. The yield a user can expect after PYR% has been applied.		
Herb\$	- the commercial name of the herbicide		
App	- the cost per acre for application		

```

5   ****
10  *          WILDOAT VERSION 1.1
15  *      Program WILDOAT predicts yield reduction due to WILD OAT
20  * infestation plus gives an economic cost vs. return analysis for
25  * herbicide use.
30  *
35  *      program written by: Michael Vasey
40  *      revised by:          Ricky M. Koon, June 1984, July 1984
45  ****
50 CLEAR: KEY OFF: SCREEN 0,0,0 : COLOR 7,0
55 ON ERROR GOTO 3295:ST$ = STRING$(79,32)
60 FKEY$ = "[F1] HELP    [F2] STOP"
65  ****
70  *          INITIALIZE FILE$
75  ****
80 NUM% = 0
85 OPEN "WILDOATS.INX" AS #1 LEN = 18
90 FIELD #1, 14 AS X.INX$, 4 AS POINT.INX$
95 GET #1
100 IF LEFT$(X.INX$,1) = "/" THEN 95
105 IF LEFT$(X.INX$,1) = "\" THEN 120
110 NUM% = NUM% + 1
115 GOTO 95
120 CLOSE
125 DIM FILE$(NUM%,7)
130 REC% = 1
135 OPEN "WILDOATS.INX" AS #1 LEN = 18
140 FIELD #1, 14 AS X.INX$, 4 AS POINT.INX$
145 OPEN "WILDOATS.DTA" AS #2 LEN = 77
150 FIELD #2,10 AS HERB$,6 AS CROP$,6 AS LOW$,6 AS HIGH$,10 AS COST$,3 AS
     RATING$,35 AS COMMENTS$
155 GET #1
160 IF LEFT$(X.INX$,1) = "/" THEN 155
165 IF LEFT$(X.INX$,1) = "\" THEN 225
170 POINTER = VAL(POINT.INX$)
175 GET #2
180 FILE$(REC%,1) = HERB$
185 FILE$(REC%,2) = CROP$
190 FILE$(REC%,3) = LOW$
195 FILE$(REC%,4) = HIGH$
200 FILE$(REC%,5) = COST$
205 FILE$(REC%,6) = RATING$
210 FILE$(REC%,7) = COMMENTS$
215 REC% = REC% + 1
220 GOTO 155
225 CLOSE
230  ****
235  *          INITIALIZE SCREENIN ROUTINE
240  ****
245 FOR J% = 1 TO 10: KEY J%,"": NEXT J%
250 DIM MINKEY(9), MAXKEY(9), EXTMINKEY(9), EXTMAXKEY(9), SPECIALKEY$(9),
     EXTSPECIALKEY$(9): TRUE = -1: FALSE = 0
255  **** caller = 1 ****
260 MINKEY(1) = 49: MAXKEY(1) = 50: EXTMINKEY(1) = 59: EXTMAXKEY(1) = 60:
     SPECIALKEY$(1) = CHR$(13)+CHR$(8)+CHR$(27)+"12": EXTSPECIALKEY$ = ""

```

```

265 **** caller = 2 ****
270 MINKEY(2) = 48: MAXKEY(2) = 57: EXTMINKEY(2) = 59: EXTMAXKEY(2) = 60:
    SPECIALKEY$(2) = CHR$(13)+CHR$(8)+CHR$(27): EXTSPECIALKEY$ = ""
275 **** caller = 3 ****
280 MINKEY(3) = 46: MAXKEY(3) = 57: EXTMINKEY(3) = 59: EXTMAXKEY(3) = 60:
    SPECIALKEY$(3) = CHR$(13)+CHR$(8)+CHR$(27): EXTSPECIALKEY$ = ""
285 **** caller = 4 ****
290 MINKEY(4) = 65: MAXKEY(4) = 122: EXTMINKEY(4) = 0: EXTMAXKEY(4) = 0:
    CAPSON(4) = -1: SPECIALKEY$(4) = CHR$(13)+CHR$(8)+CHR$(27):
    EXTSPECIALKEY$ = ""
295 **** caller = 5 ****
300 MINKEY(5) = 49: MAXKEY(5) = 53: EXTMINKEY(5) = 59: EXTMAXKEY(5) = 59:
    SPECIALKEY$(5) = CHR$(8)+CHR$(13)+CHR$(27)+"12345": EXTSPECIALKEY$(5) = ""
305 **** caller = 6 ****
310 MINKEY(6) = 49: MAXKEY(6) = 54: EXTMINKEY(6) = 59: EXTMAXKEY(6) = 59:
    SPECIALKEY$(6) = CHR$(8)+CHR$(13)+CHR$(27)+"123456": EXTSPECIALKEY$(6) = ""
315 **** caller = 7 ****
320 MINKEY(7) = 32: MAXKEY(7) = 122: EXTMINKEY(7) = 0: EXTMAXKEY(7) = 0:
    CAPSON(7) = -1: SPECIALKEY$(7) = CHR$(8)+CHR$(13)+CHR$(27):
    EXTSPECIALKEY$(7) = "": CAPSON(7) = -1
325 **** caller = 8 y/n responses ****
330 MINKEY(8) = 0: MAXKEY(8) = 0: EXTMINKEY(8) = 0: EXTMAXKEY(8) = 0:
    SPECIALKEY$(8) = CHR$(8)+CHR$(13)+CHR$(78)+CHR$(89)+CHR$(110)+CHR$(121):
    EXTSPECIALKEY$(8) = "": CAPSON(8) = -1
335 **** caller = 9 G or L response ****
340 MINKEY(9)=0: MAXKEY(9)=0: EXTMINKEY(9)=0: EXTMAXKEY(9)=0: CAPSON(9) = -1:
    SPECIALKEY$(9)= CHR$(8)+CHR$(13)+CHR$(71)+CHR$(76)+CHR$(103)+CHR$(108):
    EXTSPECIALKEY$(9) = ""
345 ****
350 *          OPENING COMMENTS          *
355 ****
360 CLS
365 LOCATE 2,12
370 PRINT "WILDOAT is designed to predict yield reduction due to wild oat"
375 PRINT TAB(12) "infestation in either wheat or barley.":PRINT
380 PRINT TAB(12) "WILDOAT will also allow you to calculate cost vs."
385 PRINT TAB(12) "return for recommended herbicide application.": PRINT
390 PRINT TAB(12) "WILDOAT was developed by Dr. E. H. Vasey, Extension"
395 PRINT TAB(12) "Soils Specialist at North Dakota State University."
400 PRINT:PRINT TAB(12) "WILDOAT utilizes data collected by the Weed
    Science Staff"
405 PRINT TAB(12)"in the Agronomy Department at NDSU"
410 PRINT:PRINT TAB(12) "Refer to current Weed Control Guide (Extension
    Circular W-253)"
415 PRINT TAB(12) "or current Crop Production Guide for more complete
    information"
420 PRINT TAB(12) "on wild oat control in wheat or barley."
425 LOCATE 25,21,0:COLOR 0,7:PRINT " press any key to continue ";:COLOR 7,0
430 A$=INKEY$:IF A$=""THEN 430
435 ****
440 *          GET CROP CHOICES          *
445 ****
450 CLS
455 LOCATE 3,12:PRINT "CROP CHOICES ARE:"
460 LOCATE 5,15:PRINT "1. WHEAT"
465 LOCATE 6,15:PRINT "2. BARLEY"

```

```

470 LOCATE 8,12:PRINT "Enter your choice of crop (by number)"
475 LOCATE 25,12:PRINT FKEY$;
480 CALLER = 1: XFLD = 50: YFLD = 8: LNGTH = 2
485 GOSUB 2630
490 IF E$ <> "" THEN 495 ELSE 510
495 C = ASC(E$)
500 IF C = 59 THEN GOSUB 1775 : GOTO 450
505 IF C = 60 THEN 1470
510 CROP = VAL(S$)
515 ON CROP GOTO 520,555
520 ****
525 /* * WHEAT SECTION * */
530 ****
535 CLS: LOCATE 25,12:PRINT FKEY$;:PRINT:PRINT:PRINT: CROP$ = "WHEAT"
540 GOSUB 710
545 GOSUB 880
550 GOTO 1430
555 ****
560 /* * BARLEY SECTION * */
565 ****
570 CLS: LOCATE 25,12:PRINT FKEY$;:PRINT:PRINT:PRINT: CROP$ = "BARLEY"
575 GOSUB 1475
580 GOSUB 880
585 GOTO 1430
590 ****
595 /* * HELP FOR ESTIMATING WILD OAT INFESTATION * */
600 ****
605 CLS: LOCATE 3,12,0
610 PRINT " A good estimate can be obtained by counting the"
615 PRINT TAB(12) "wild oat plants in several one square yard samples"
620 PRINT TAB(12)"and finding the average. Sorry, that is the only way!"
625 PRINT TAB(12) "A very rough estimate might be 10 for a mild infestation,"
630 PRINT TAB(12)"40 for a moderate, and over 80 for severe."
635 PRINT
640 LOCATE 25,21: PRINT "press any key to continue";
645 A$=INKEY$: IF A$ = "" THEN 645
650 DEF SEG:CLS
655 RETURN
660 ****
665 /* * TYPICAL CUSTOM APPLICATION RATES FOR 1984 * */
670 ****
675 LOCATE 16,1,0:PRINT:COLOR 0,7:PRINT TAB(6) "Typical custom rates for
1984 are: "
680 PRINT TAB(8) "$2.00 per acre for ground application"
685 PRINT TAB(8) "$3.00 per acre for aerial application"
690 PRINT TAB(8) "$2.00 per acre for incorporation pass";
695 COLOR 7,0:FOR I = 1 TO 3500:NEXT I
700 LOCATE 16,1,0: FOR I=1 TO 5: PRINT STRING$(50,32): NEXT I
705 RETURN
710 ****
715 /* * CALCULATE YIELD REDUCTION FOR WHEAT * */
720 ****
725 LOCATE 3,12:PRINT "What is your PROJECTED yield of ";CROP$;" in bu/a?";
730 CALLER = 2: XFLD = POS(0)+1: YFLD = 3: LNGTH = 3
735 GOSUB 2630
740 IF E$ <> "" THEN 745 ELSE 760

```

```

745 C = ASC(E$)
750 IF C = 59 THEN GOSUB 1915: GOTO 725
755 IF C = 60 THEN 1470
760 YIELD = VAL(S$)
765 LOCATE 5,12:PRINT "What is the number of WILD OATS per square yard"
770 PRINT TAB(12) "(if the number exceeds 160 then use 160)?";
775 CALLER = 2: XFLD = POS(0) + 1: YFLD = 6: LNGTH = 4
780 GOSUB 2630
785 IF E$ <> "" THEN 790 ELSE 805
790 C = ASC(E$)
795 IF C = 59 THEN GOSUB 605: GOTO 765
800 IF C = 60 THEN 1470
805 IF VAL(S$) > 286 THEN NUMBER = 286 ELSE NUMBER = VAL(S$)
810 NUMBER$ = S$
815 PYR% = .0000022*(NUMBER^3)+2.89*SQR(NUMBER) 'pyr is percent yield reduction
820 YR = YIELD-(PYR%/100*YIELD) ' yr is yield reduction
825 GOSUB 835
830 RETURN
835 ****
840 /* PRINT RESULTS OF CALCULATIONS */
845 ****
850 CLS: LOCATE 3,12
855 PRINT "An infestation of ";NUMBER$;" wild oats per square yard"
860 PRINT TAB(12) "can reduce your yield by";PYR%;" percent.":PRINT
865 PRINT TAB(12) "Your yield may be reduced from";YIELD;" bu/a"
870 PRINT TAB(12) "to ";:PRINT USING"##.#";YR;:PRINT " bu/a unless wild
oats are controlled."
875 RETURN
880 ****
885 /* HERBICIDE APPLICATION ROUTINE */
890 ****
895 LOCATE 25,1:PRINT ST$;:LOCATE 23,6,0
900 PRINT "Do you wish to see a cost vs. return for herbicide application(Y/N)";
905 CALLER = 8: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
910 GOSUB 2720
915 IF S$<>"Y" AND S$<>"y" THEN 1375
920 CLS
925 PRINT "The recommended herbicides for ";CROP$;" are:"
930 PRINT
935 PRINT TAB(15) "APPLICATION" : PRINT TAB(17) "RATES/A"
940 PRINT TAB(2) "HERBICIDE" TAB(15)"LOW" TAB(22)"HIGH" TAB(31)"COST"
TAB(39)"RATING" TAB(59)"COMMENTS"
945 PRINT TAB(2) "-----" TAB(15) "---" TAB(22) "----" TAB(31) "----"
TAB(39) "-----" TAB(59) "-----"
950 LNE% = 1
955 FOR I = 1 TO NUM%
960 IF LEFT$(FILE$(I,2),LEN(CROP$)) = CROP$ THEN PRINT LNE%;FILE$(I,1)
TAB(15)FILE$(I,3) TAB(22)FILE$(I,4) TAB(29)FILE$(I,5) TAB(41)FILE$(I,6)
TAB(45)FILE$(I,7): LNE% = LNE% + 1
965 NEXT I
970 COLOR 0,7:PRINT" RATINGS: G = GOOD F = FAIR P = POOR N = NONE
S = SELDOM O = OFTEN ":COLOR 7,0
975 GOSUB 2400
980 LOCATE 25,12: PRINT FKEY$;
985 LOCATE 15,1:PRINT ST$: LOCATE 15,1:PRINT "2. What is your cost per
acre for application of ";HERB$;

```

```

990 CALLER = 3: XFLD = POS(0)+1: YFLD = 15: LNGTH = 5
995 GOSUB 2630
1000 IF E$ <> "" THEN 1005 ELSE 1020
1005 C = ASC(E$)
1010 IF C = 59 THEN GOSUB 675: GOTO 985
1015 IF C = 60 THEN 1470
1020 APP = VAL(S$)
1025 LOCATE 15,1:PRINT ST$:LOCATE 15,1:PRINT "3. What will you pay per
gallon or per pound for ";HERB$;
1030 CALLER = 3: XFLD = POS(0)+1: YFLD = 15: LNGTH = 6
1035 GOSUB 2630
1040 IF E$ <> "" THEN 1045 ELSE 1060
1045 C = ASC(E$)
1050 IF C = 59 THEN GOSUB 1830: IF PR=0 THEN 1025 ELSE PRICE = PR:GOTO 1065
1055 IF C = 60 THEN 1470
1060 PRICE = VAL(S$)
1065 LOCATE 15,1:PRINT ST$;:LOCATE 15,1:PRINT "4. What rate in pints/A
or pounds/A will you apply ";HERB$;
1070 PRINT " (INAPPROPRIATE RATES WILL YIELD FALSE ANALYSIS!);"
1075 CALLER = 3: XFLD = POS(0)+1: YFLD = 16: LNGTH = 5
1080 GOSUB 2630
1085 IF E$ <> "" THEN 1090 ELSE 1105
1090 C = ASC(E$)
1095 IF C = 59 THEN GOSUB 1960 : GOTO 1065
1100 IF C = 60 THEN 1470
1105 RATE = VAL(S$)
1110 LOCATE 25,1: PRINT ST$;
1115 LOCATE 15,1:PRINT STRING$(160," "):LOCATE 15,1:PRINT "5. What is
your expected selling price per bushel for ";CROP$;
1120 CALLER = 3: XFLD = POS(0)+1: YFLD = 15: LNGTH = 5
1125 GOSUB 2630
1130 IF E$ <> "" THEN 1135 ELSE 1150
1135 C = ASC(E$)
1140 IF C = 59 THEN GOSUB 1830: GOTO 1115
1145 IF C = 60 THEN 1470
1150 SELL = VAL(S$)
1155 LOCATE 25,1: PRINT STRING$(79,32);
1160 LOCATE 15,1: PRINT ST$: LOCATE 15,1: PRINT "1. Herbicide: ";HERB$;
1165 LOCATE 16,1:PRINT ST$: LOCATE 16,1:PRINT "2. Cost/A for Application:
";:PRINT USING "$$##.##";APP
1170 LOCATE 17,1:PRINT ST$:LOCATE 17,1:PRINT "3. Price: ";:PRINT USING
"$$##.##";PRICE
1175 LOCATE 18,1:PRINT ST$:LOCATE 18,1:PRINT "4. Rate of Application in
Pints/A or Pounds/A: ";:PRINT USING "#.##";RATE
1180 LOCATE 19,1: PRINT ST$:LOCATE 19,1:PRINT "5. Selling Price:";:PRINT
USING "$$##.##";SELL
1185 PRINT:COLOR 0,7:PRINT" DATA CORRECT (Y/N)? ";:COLOR 7,0
1190 CALLER = 8: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
1195 GOSUB 2720
1200 IF S$ = "n" OR S$ = "N" THEN PRINT "Which question number contains
the error"; ELSE 1215
1205 INPUT ER:LOCATE 22,1:PRINT ST$: ON ER GOSUB 2175,2200,2235,2265,2305
1210 GOTO 1160
1215 CLS
1220 IF HERB$ = "FAR-GO 10G" OR LORG$ = "G" THEN GOSUB 1380 ELSE GOSUB 1405
1225 LOCATE 3,6

```

```

1230 PRINT "Your price/A for ";HERB$;" and application is";
1235 PRINT USING "$$###.##";TAL
1240 PRINT: NCR = YR*SELL ' ncr is no control of wildoats expected return
1245 PRINT TAB(6) "Your gross return without control would be: ";
1250 PRINT USING "$$###.##";NCR:PRINT
1255 CR = YIELD * SELL
1260 PRINT TAB(6) "Your gross return with optimum control (100%) would
be:";:PRINT USING "$$###.##";CR
1265 PRINT: PRINT TAB(6) "The return of optimum control above no control is";
1270 PRINT USING "$$###.##";CR-NCR
1275 PRINT: PRINT TAB(6) "Your investment for control was";:PRINT USING
"$$###.##";TAL;:PRINT " this"
1280 PRINT TAB(6)"gives a return of";:PRINT USING "$$###.##";(CR-NCR)/TAL;
:PRINT " per dollar invested."
1285 PRINT: D = CR-NCR
1290 IF D < TAL THEN 1315
1295 PRINT TAB(6) "Economic analysis indicates that Wild Oat control"
1300 PRINT TAB(6) "would be economical in the short term."
1305 PRINT TAB(6) "Added long term benefits make control even more attractive."
1310 GOTO 1335
1315 PRINT TAB(6) "Economic analysis indicates that Wild Oat control would"
1320 PRINT TAB(6) "not be economical in the short term."
1325 PRINT TAB(6) "However added long term benefits may still make control"
1330 PRINT TAB(6) "profitable."
1335 LOCATE 23,10,0:COLOR 0,7:PRINT " Would you like a printout of the
compiled results (Y/N)?";:COLOR 7,0
1340 CALLER = 8: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
1345 GOSUB 2720
1350 IF S$="Y" OR S$="y" THEN GOSUB 2030
1355 LOCATE 23,1:PRINT ST$;:LOCATE 23,6,0:PRINT "Would you like to compare
cost vs return for another herbicide(Y/N)?";
1360 CALLER = 8: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
1365 GOSUB 2720
1370 IF S$ = "Y" OR S$ = "y" THEN IF CROP$="BARLEY" THEN 2335 ELSE 920
1375 RETURN
1380 ****
1385 '* CALCULATE GRANULAR COST *
1390 ****
1395 TAL = RATE * PRICE + APP
1400 RETURN
1405 ****
1410 '* CALCULATE LIQUID COST *
1415 ****
1420 TAL = (RATE * (PRICE/8)) + APP
1425 RETURN
1430 ****
1435 '* RETURN PERFORMANCE *
1440 ****
1445 CLS: LOCATE 3,12
1450 PRINT "Would you like to see another prediction (Y/N)?";
1455 CALLER = 8: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
1460 GOSUB 2720
1465 IF S$="Y" OR S$="y" THEN 450
1470 CLS: SYSTEM

```

```

1475 ****
1480 /* CALCULATE YIELD REDUCTION FOR BARLEY */
1485 ****
1490 LOCATE 3,12:PRINT "What is your projected yield of ";CROP$;" in bu/a";
1495 CALLER = 3: XFLD = POS(0)+1: YFLD = 3: LNGTH = 3
1500 GOSUB 2630
1505 IF E$ <> "" THEN 1510 ELSE 1525
1510 C = ASC(E$)
1515 IF C = 59 THEN GOSUB 1915: GOTO 1490
1520 IF C = 60 THEN 1470
1525 YIELD = VAL(S$)
1530 LOCATE 25,1: PRINT ST$;
1535 LOCATE 5,12:PRINT "Is your ";CROP$;" on fertilized (f) or non-fertilized
  (nf) soil"
1540 PRINT TAB(12) "(fertilized consists of adequate rates of Nitrogen"
1545 PRINT TAB(12) "and Phosphorus)";
1550 CALLER = 4: XFLD = POS(0)+1: YFLD = 7: LNGTH = 3
1555 GOSUB 2630
1560 IF E$ <> "" THEN 1565 ELSE 1580
1565 C = ASC(E$)
1570 IF C = 59 THEN GOSUB 1830: GOTO 1535
1575 IF C = 60 THEN 1470
1580 FERT$ = S$
1585 IF FERT$<>"NF" AND FERT$<>"nf" AND FERT$<>"F" AND FERT$<>"f" THEN BEEP:
  GOTO 1535
1590 LOCATE 25,12: PRINT FKEY$;
1595 IF FERT$="NF" OR FERT$="nf" THEN GOSUB 1700 ELSE GOSUB 1610
1600 GOSUB 835
1605 RETURN
1610 ****
1615 /* YIELD REDUCTION ON FERTILIZED SOILS */
1620 ****
1625 LOCATE 9,12:PRINT "What is the number of wild oat plants per square yard"
1630 PRINT TAB(12) "(if the number exceeds 250 then use 250)";
1635 CALLER = 2: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 4
1640 GOSUB 2630
1645 IF E$ <> "" THEN 1650 ELSE 1665
1650 C = ASC(E$)
1655 IF C = 59 THEN GOSUB 605: GOTO 1625
1660 IF C = 60 THEN 1470
1665 NUMBER$ = S$
1670 O=VAL(NUMBER$)^3      # wild oats cubed
1675 W=VAL(NUMBER$)*VAL(NUMBER$)
1680 A=.0023244 * W : B=-7.24E-06 * 0
1685 PYR% = A + B
1690 YR = YIELD - (PYR%/100 * YIELD)
1695 RETURN
1700 ****
1705 /* YIELD REDUCTION ON NON-FERTILIZED SOILS */
1710 ****
1715 LOCATE 9,12:PRINT "What is the number of wild oat plants per square yard"
1720 PRINT TAB(12) "(if the number exceeds 250 then use 250)";
1725 CALLER = 2: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 4
1730 GOSUB 2630
1735 IF E$ <> "" THEN 1740 ELSE 1755

```

```

1740 C = ASC(E$)
1745 IF C = 59 THEN GOSUB 605: GOTO 1715
1750 IF C = 60 THEN 1470
1755 NUMBER$ = S$
1760 PYR% = 2.95573 * (SQR(VAL(NUMBER$)))
1765 YR = YIELD - (PYR%/100 * YIELD)
1770 RETURN
1775 ****
1780 /* HELP FOR CROP SELECTION */
1785 ****
1790 CLS: LOCATE 3,12:PRINT "You are currently limited to either Wheat or
Barley."
1795 PRINT TAB(7) "Further crop selections are planned for future versions
of WILDOAT."
1800 LOCATE 25,21,0:COLOR 0,7:PRINT " press any key to continue ";:COLOR 7,0
1805 A$ = INKEY$: IF A$ = "" THEN 1805
1810 LOCATE 25,12: PRINT FKEY$;: RETURN
1815 ****
1820 /* HELP FOR PRICE/GAL OR PRICE/POUND */
1825 ****
1830 LOCATE 17,10,0:COLOR 0,7:PRINT " The above table provides recommended
prices/gallon/lb ":COLOR 7,0:FOR J% = 1 TO 3000: NEXT J%:LOCATE 17,1:
PRINT ST$;
1835 LOCATE 17,5,0:PRINT "Do you need to calculate a per gallon or per
pound price(Y/N)?";
1840 CALLER = 8: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
1845 GOSUB 2720
1850 IF S$ <> "y" AND S$ <> "Y" THEN 1910
1855 LOCATE 17,5,0:PRINT "What size is the container you will use
(i.e. 5 gal, 30 gal, 100 lb)?";
1860 CALLER = 2: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 4
1865 GOSUB 2630
1870 SIZE = VAL(S$)
1875 LOCATE 17,1:PRINT ST$;:LOCATE 17,5:PRINT "What is the price of the
container?";
1880 CALLER = 3: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 8
1885 GOSUB 2630
1890 LOCATE 17,1:PRINT ST$;:PRCE = VAL(S$)
1895 PR = PRCE/SIZE
1900 LOCATE 17,1:PRINT ST$;:LOCATE 17,5:PRINT "The price is";:PRINT USING
"$$##.#";PR;:PRINT " and will be automatically placed in the program"
1905 FOR J% = 1 TO 3000: NEXT J%
1910 LOCATE 17,1:PRINT ST$;:RETURN
1915 ****
1920 /* HELP FOR PROJECTED YIELD */
1925 ****
1930 CLS: LOCATE 3,12: PRINT "Yield estimates can be your own estimates
based on"
1935 PRINT TAB(7) "experience or they can be based on variety performance
data from branch"
1940 PRINT TAB(7) "experiment stations. They can also be based on soil
survey reports."
1945 LOCATE 25,21,0:COLOR 0,7:PRINT " press any key to continue ";:COLOR 7,0
1950 A$ = INKEY$: IF A$ = "" THEN 1950
1955 CLS:LOCATE 25,12: PRINT FKEY$;: RETURN

```

```

1960 ****
1965 '*          HELP FOR RATE/A OR POUND/A      *
1970 ****
1975 LOCATE 18,12,0: PRINT ``RATES ARE BASED on broadcast application and are"
1980 PRINT TAB(7) "expressed as active ingredient or acid equivalent, and
as the"
1985 PRINT TAB(7) "amount of commercial product. Commercial formulations
of the"
1990 PRINT TAB(7) "same herbicide may vary in their amount of active ingredient."
1995 PRINT TAB(7) "The table above provides recommended low/high application
rates/A."
2000 FOR J% = 1 TO 10000: NEXT J%
2005 LOCATE 18,1: FOR J% = 1 TO 5: PRINT ST$: NEXT J%
2010 LOCATE 25,12,0: PRINT FKEY$;
2015 RETURN
2020 ****
2025 '*          PRINT SUBROUTINE      *
2030 ****
2035 LPRINT CHR$(27);CHR$(53)
2040 IF CROP$="BARLEY" THEN IF FERT$="F" OR FERT$="f" THEN LPRINT CHR$(14);"
FERTILIZED "; ELSE LPRINT CHR$(14); "      NON-FERTILIZED ";
2045 LPRINT CHR$(14);CROP$;" GUIDELINES"
2050 LPRINT:LPRINT DATE$" at "TIME$: LPRINT:LPRINT
2055 LPRINT "Projected yield:";YIELD; TAB(40) "Wild Oats/sq yard: ";NUMBER$
2060 LPRINT:LPRINT "Yield with no control: ";:LPRINT USING"##.#";YR
2065 LPRINT:LPRINT CHR$(14); "Herbicide: ";HERB$
2070 LPRINT: LPRINT "Cost/A for Application:";:LPRINT USING "$$##.##";APP
2075 LPRINT:LPRINT "Price:";:LPRINT USING "$$##.##";PRICE;:LPRINT TAB(40)
"Rate of Application(units/A):";RATE
2080 LPRINT:LPRINT "Expected market price:";:LPRINT USING "$$##.##";SELL:LPRINT
2085 LPRINT "Price/A and application:";:LPRINT USING "$$##.##";TAL:LPRINT
2090 LPRINT "Gross return without control:";:LPRINT USING "$$##.##";NCR:LPRINT
2095 LPRINT "Gross return with optimum control:";:LPRINT USING "$$##.##";
CR:LPRINT
2100 LPRINT "Return of optimum control above no control is:";:LPRINT USING
"$$##.##";CR-NCR
2105 LPRINT:LPRINT CHR$(14); "Your investment return is:";:LPRINT CHR$(14);:LPRINT
USING "$$##.##";(CR-NCR)/TAL;:LPRINT CHR$(14);" per dollar invested."
2110 IF D < TAL THEN 2135
2115 LPRINT: LPRINT "Economic analysis indicates that Wild Oat control would"
2120 LPRINT "be economical in the short term."
2125 LPRINT "Added long term benefits make control even more attractive."
2130 GOTO 2150
2135 LPRINT:LPRINT "Economic analysis indicates that Wild Oat control would"
2140 LPRINT "NOT be economical in the short term."
2145 LPRINT "However added long term benefits may still make control
profitable."
2150 LPRINT CHR$(12)
2155 RETURN
2160 ****
2165 '*          CORRECT INPUT DATA ROUTINE FOR LINES 1300-1440      *
2170 ****
2175 ' get the correct herbicide
2180 LOCATE 22,1:GOSUB 2435
2185 LOCATE 22,1: PRINT ST$

```

```

2190 RETURN
2195 ' get correct cost/A for application
2200 LOCATE 22,1:PRINT "What is your cost per acre for application";
2205 CALLER = 3: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 5
2210 GOSUB 2630
2215 APP = VAL($$)
2220 LOCATE 22,1 : PRINT ST$
2225 RETURN
2230 ' get correct price/gal or price/lb
2235 LOCATE 22,1:PRINT "What will you pay per unit of measure for ";HERB$;
2240 CALLER = 3: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 6
2245 GOSUB 2630
2250 PRICE = VAL($$)
2255 LOCATE 22,1: PRINT ST$
2260 RETURN
2265 LOCATE 22,1:PRINT "What rate in units/A will you apply(INACCURACY
MEANS FALSE RESULTS!)" ;
2270 ' get the correct application rate
2275 CALLER = 3: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 5
2280 GOSUB 2630
2285 RATE = VAL($$)
2290 LOCATE 22,1: PRINT ST$
2295 RETURN
2300 ' get the correct expected selling price
2305 LOCATE 22,1:SELL=0:PRINT "What is your expected selling price per
bushel for ";CROP$;
2310 CALLER = 3: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 5
2315 GOSUB 2630
2320 SELL = VAL($$)
2325 LOCATE 22,1: PRINT STRING$(79," ")
2330 RETURN
2335 ****
2340 '* REVERSE STATED FERTILIZATION *
2345 ****
2350 CLS:LOCATE 3,6:PRINT "Your present calculations have been on ";
2355 IF FERT$="f" OR FERT$="F" THEN PRINT "fertilized" ELSE PRINT
"non-fertilized"
2360 PRINT TAB(6) "soil. Do you wish to continue with the present
option(y/n)?";
2365 CALLER = 8: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
2370 GOSUB 2720
2375 IF S$="Y" OR S$="y" THEN 2395
2380 IF FERT$="f" OR FERT$="F" THEN GOSUB 1760 ELSE GOSUB 1670
2385 IF FERT$="F" OR FERT$="f" THEN FERT$="nf" ELSE FERT$="F"
2390 GOSUB 835
2395 GOTO 895
2400 ****
2405 '* HERBICIDE SELECTION ROUTINE *
2410 '** Lets the user chose a herbicide by number and sets herb$ to the *
2415 ** correct herbicide name. *
2420 ****
2425 LOCATE 15,1:PRINT ST$:LOCATE 25,12:PRINT "[F1] Own herbicide";
2430 LOCATE 15,1:PRINT "1. ";
2435 PRINT "Choose by number the herbicide you will apply([F1] for own)";
2440 IF CROP$ = "BARLEY" THEN 2505
2445 CALLER = 5: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2

```

```

2450 GOSUB 2630
2455 IF E$ <> "" THEN 2460 ELSE 2470
2460 C = ASC(E$)
2465 IF C = 59 THEN GOTO 2575
2470 HN = VAL( $$ )
2475 IF HN = 1 THEN HERB$ = "AVENGE"
2480 IF HN = 2 THEN HERB$ = "CARBYNE"
2485 IF HN = 3 THEN HERB$ = "FAR-GO 10G"
2490 IF HN = 4 THEN HERB$ = "FAR-GO 4L"
2495 IF HN = 5 THEN HERB$ = "HOELON"
2500 GOTO 2570
2505 ' herbicides for barley
2510 CALLER = 6: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
2515 GOSUB 2630
2520 IF E$ <> "" THEN 2525 ELSE 2535
2525 C = ASC(E$)
2530 IF C = 59 THEN GOTO 2575
2535 HN = VAL( $$ )
2540 IF HN = 1 THEN HERB$ = "AVADEX(L)"
2545 IF HN = 2 THEN HERB$ = "AVENGE"
2550 IF HN = 3 THEN HERB$ = "CARBYNE"
2555 IF HN = 4 THEN HERB$ = "FAR-GO 10G"
2560 IF HN = 5 THEN HERB$ = "FAR-GO 4L"
2565 IF HN = 6 THEN HERB$ = "HOELON"
2570 RETURN
2575 '-----> allow user input own herbicide
2580 LOCATE 25,1: PRINT ST$;:LOCATE 15,1:PRINT ST$;:LOCATE 15,5:PRINT
  "What is the name of the herbicide you will use?";
2585 CALLER = 7:XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 16
2590 GOSUB 2630
2595 HERB$ = $$ 
2600 LOCATE 17,5:PRINT "Is the herbicide liquid(L) or granular(G)?";
2605 CALLER = 9: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
2610 GOSUB 2630
2615 LORG$ = $$ 
2620 LOCATE 17,1: PRINT STRING$ (75,32)
2625 GOTO 2570
2630 '-----> screenin routine
2635 '-----> taken from `Advanced Basic and Beyond` by Dr. L. Goldstein
2640 '-----> this routine inputs data as a string $$ 
2645 'It allows input to have the following parameters:
2650 'LNGTH = maximum length of input string
2655 'XFLD = cursor column for beginning of input string
2660 'YFLD = cursor row for input field
2665 'FLDBEG = first character position in field
2670 'FLDEND = last character position in field
2675 ' $$ = contents of the field from beginning up to space before cursor
2680 'T$ = contents of the field from the cursor to the end of the field
2685 'At the end of the routine , the contents of the field are returned in $$ 
2690 'LASTPOS = position currently occupied by last character
2695 'If a key with an extended ASCII code is pressed, it ends processing the
2700 'current field
2705 'The contents of the field are returned in $$, the second byte of the
2710 'extended ASCII code in E$.
2715 'CSR = the current column of the cursor
2720 '-----> main routine

```

```

2725 S$ = "": E$ = "": INPUTEND = FALSE: KEYHIT = FALSE
2730 FLDEND = XFLD + (LNGTH-1)
2735 CSR = XFLD
2740 GOSUB 3060      '-----> compute initial lastpos
2745 LOCATE YFLD, XFLD: COLOR 0,7:PRINT SPC(LNGTH-1):COLOR 7,0:LOCATE YFLD,XFLD
2750 WHILE INPUTEND = FALSE
2755   GOSUB 3140      '-----> input character
2760   IF EXTENDED = TRUE THEN 2830 ELSE 2805
2765 WEND
2770 J% = (XFLD+LNGTH) - LASTPOS
2775 PRINT STRING$ (J%,32)
2780 GOSUB 2980      '-----> read screen
2785 S$ = S$ + T$
2790 RETURN
2795 '-----> begin subroutines <-----
2800 '-----> handle ordinary ASCII codes
2805 KEYHIT = TRUE
2810 IF C$ = CHR$(8) THEN 2870  'Backspace
2815 IF C$ = CHR$(13) THEN 2905  'Enter
2820 IF C$ = CHR$(27) THEN 2920  'Esc
2825 IF C$ >= CHR$(32) THEN GOTO 2950  'handle displayable character
2830 '-----> handle extended ASCII codes
2835 E$ = C$
2840 INPUTEND = TRUE
2845 GOTO 2765
2850 '-----> reject character
2855 BEEP
2860 GOTO 2765
2865 '-----> handle backspace
2870 IF LASTPOS < XFLD THEN 2855
2875 GOSUB 2980
2880 IF CSR <> XFLD THEN CSR = CSR-1:PRINT CHR$(219);
2885 LOCATE YFLD,CSR:PRINT CHR$(219);:LOCATE YFLD,CSR
2890 LASTPOS = LASTPOS-1
2895 GOTO 2765
2900 '-----> handle enter
2905 INPUTEND = TRUE
2910 GOTO 2765
2915 '-----> handle esc
2920 LOCATE YFLD,XFLD
2925 PRINT STRING$(LNGTH,32);
2930 LASTPOS = 0: CSR = XFLD
2935 LOCATE YFLD, XFLD:COLOR 0,7:PRINT SPC(LNGTH-1):COLOR 7,0:LOCATE YFLD,XFLD
2940 GOTO 2765
2945 '-----> display character
2950 PRINT C$;
2955 IF LASTPOS < CSR THEN LASTPOS = CSR
2960 IF CSR = FLDEND THEN BEEP
2965 IF CSR = FLDEND THEN PRINT CHR$(29); CHR$(32); CHR$(29);ELSE CSR = CSR + 1
2970 GOTO 2765
2975 '-----> read field from screen
2980 LOCATE,,0
2985 S$ = "": T$ = ""
2990 IF LASTPOS = 0 THEN 3030
2995 FOR J% = XFLD TO CSR-1
3000   S$ = S$ + CHR$(SCREEN(YFLD,J%))

```

```

3005 NEXT J%
3010 FOR J% = CSR TO LASTPOS
3015   T$ = T$ + CHR$(SCREEN(YFLD,J%))
3020 NEXT J%
3025 LOCATE,,1
3030 RETURN
3035 '-----> erase field
3040 LOCATE YFLD, XFLD: CSR = XFLD: LASTPOS = 0
3045 PRINT STRING$(LNGTH,32);
3050 LOCATE YFLD, XFLD
3055 RETURN
3060 '-----> compute lastpos
3065 LASTPOS = FLDEND: CSR = XFLD
3070 GOSUB 2980
3075 WHILE RIGHT$(T$,1) = CHR$(32)
3080   T$ = LEFT$(T$,LEN(T$)-1)
3085   LASTPOS = LASTPOS -1
3090 WEND
3095 RETURN
3100 '-----> clear keyboard buffer
3105 DEF SEG = 0: POKE 1050,PEEK(1052)
3110 DEF SEG
3115 RETURN
3120 '-----> end routines
3125 '-----> keyin
3130 'This routine reads a character from the keyboard and accepts or
3135 'rejects it based on the caller's specifications
3140 '-----> input character string from inkey$
3145 C$ = INKEY$
3150 IF C$ = "" THEN 3145
3155 C = ASC(C$)
3160 IF LEN(C$) = 2 THEN EXTENDED = TRUE ELSE EXTENDED = FALSE
3165 IF EXTENDED = FALSE THEN 3195
3170 C$ = RIGHT$(C$,1)
3175 C = ASC(C$)
3180 GOTO 3250
3185 '-----> ordinary ASCII codes
3190 '----> test for range
3195 IF C >= MINKEY(CALLER) AND C <= MAXKEY(CALLER) THEN 3220
3200 '-----> handle special characters
3205 IF SPECIALKEY$(CALLER) = "" THEN 3280
3210 IF INSTR(SPECIALKEY$(CALLER),C$) = 0 THEN 3280
3215 '-----> convert to capitals if necessary
3220 IF CAPSON(CALLER) = FALSE THEN 3290
3225 IF C > 96 AND C < 123 THEN C = C-32
3230 C$ = CHR$(C)
3235 GOTO 3290
3240 '-----> extended ASCII codes
3245 '----> test for range
3250 IF C >= EXTMINKEY(CALLER) AND C <= EXTMAXKEY(CALLER) THEN 3290
3255 '-----> handle special characters
3260 IF EXTSPECIALKEY$(CALLER) = "" THEN 3280
3265 IF INSTR(EXTSPECIALKEY$(CALLER),C$) = 0 THEN 3280
3270 GOTO 3290
3275 '-----> illegal character
3280 BEEP

```

```
3285 GOTO 3145
3290 RETURN
3295 '-----> error trapping
3300 IF ERR = 25 THEN 3310 ELSE 3330
3305 '-----> print message printer not on
3310 CLS: LOCATE 3,20: PRINT "Your printer is not on!"
3315 PRINT TAB(10) "Make sure to align your forms to top-of-page"
3320 FOR I% = 1 TO 5000: NEXT I% :CLS:RESUME 1225
3325 '-----> print unrecoverable error message
3330 CLS: LOCATE 3,10: PRINT "An error has occurred in the program that is not"
3335 PRINT TAB(10) "recoverable. Type 'RUN' to restart the program."
3340 CLS: END
```

Cooperative Extension Service, North Dakota State University of Agriculture and Applied Science, and U. S. Department of Agriculture cooperating. Myron D. Johnsrud, Director, Fargo, North Dakota. Distributed in furtherance of the Acts of Congress of May 8 and June 30, 1914. We offer our programs and facilities to all persons regardless of race, color, sex, religion, age, national origin, or handicap; and are an equal opportunity employer.